

1. IDENTIFICATION
  - 1.1 PDP-8 Maindec 826-B
  - 1.2 680 5-Bit Character Exerciser
  - 1.3 November 17, 1965



## 2. ABSTRACT

The 680 5-Bit Character Exerciser program further verifies correct operation of the 680 Data Communications System. This test assumes that the Teletype lines are full duplex. However, if the line outputs are jumpered to the line inputs, the test verifies that the input characters are received as transmitted.

## 3. REQUIREMENTS

### 3.1 Storage

#### 3.1.1 Exerciser Program

Address 0

Concurrent Output Test	0200 <sub>8</sub> to 0377 <sub>8</sub>
Output Data Table	0400 <sub>8</sub> to 0535 <sub>8</sub>
Concurrent Input Test	0600 <sub>8</sub> to 0616 <sub>8</sub>
Output Switch Register Routine	1000 <sub>8</sub> to 1074 <sub>8</sub>
Output Buffer	2200 to 2377 <sub>8</sub>
Input Buffer	2400 to 2577 <sub>8</sub>
Interrupt Routine	0001 to 0004

#### 3.1.2 Character Assembly Subroutines (List of Items)

TT5BGN	Start of Subroutines	5200 <sub>8</sub> to 5577 <sub>8</sub>
T5IN	TTI Area	5600 <sub>8</sub> to 6400 <sub>8</sub>
T5IBF	Input Buffer	7200 <sub>8</sub> to 7577 <sub>8</sub>
T5OBF	Output Buffer	6600 <sub>8</sub> to 6777 <sub>8</sub>
T5OBF2	Second Output Buffer	7000 <sub>8</sub> to 7177 <sub>8</sub>
TT5PG0	Page 0 Constants	0145 <sub>8</sub> to 0176 <sub>8</sub>
TTCHAR	Character Register	0177 <sub>8</sub>
T5AX1	Autoindex	10
T5AX2	Autoindex	11
T5AX3	Autoindex	12
T5AX4	Autoindex	13

### 3.2 Subprograms and/or Subroutines

Digital-8-35-S-A  
680 5-bit Character Assembly Subroutines

### 3.3 Equipment

Minimum Configuration PDP-8  
680 DCS hardware (Including a 50 or 75 Baud Clock)

3.4 Miscellaneous

The 50 or 75 Baud Clock in clock 1.

4. USAGE

4.1 Loading

4.1.1 If the Binary loader beginning at address 7777 is in memory, go to Section 4.1.2. Otherwise the RIM Loader and/or the Binary Loader must be loaded into memory.

4.1.2 The 6805-Bit Character Exerciser may now be loaded into memory as follows:

Set  $7777_8$  in the SWITCH REGISTER

Press LOAD ADDRESS

Place the Character Exerciser in the keyboard reader

Press START

Engage the keyboard reader

4.3 Switch Settings (See Section 4.4)

4.4 Start up and/or Entry

4.4.1 Concurrent Output Test

Set the SWITCH REGISTER to 0200

Press the LOAD ADDRESS key

Set the SWITCH REGISTER equal to the lowest line number to be tested.

Press START; the processor halts at address 0202.

Set the SWITCH REGISTER equal to the number of lines to be tested.

Press CONTINUE; the program outputs a fixed message on all the lines selected.

4.4.2 Concurrent Input Test

Set the SWITCH REGISTER to 0600.

Press the LOAD ADDRESS key.

Press START; the program scans 128 lines for input and retransmits characters on the same line as the one in which they are received.

Output SWITCH REGISTER test

Set the SWITCH REGISTER to 1000.

Press the LOAD ADDRESS key.

Set the SWITCH REGISTER to the line number to be tested.

Press START; the processor halts at address 1002.

Set SWITCH REGISTER bits 7 through 11 equal to the character to be transmitted, and SW0 as follows:

SW0-down: Do not test for input

SW0-up: Wait for input and verify that it is the same character as is transmitted. (See the end of Section 4.4.1.)

Press CONTINUE; the program transmits a carriage return-line feed, the contents of AC switches 64 times, and then repeats.

Note that since the input is two characters behind the output, (due to double buffering of the Character Assembly Subroutines) SW0 may not be changed from 0 to a 1 once the program is running. It may, however, be changed from a 1 to a 0 and switches 7 through 11 may be altered at any time.

#### 4.5 Errors in Usage

Hardware malfunctions detected by the program result in a processor halt. The following is a list of these error halts and their meanings:

##### 4.5.1 Address 0002 Not Clock Interrupt

Either an interrupt was received from a device other than Teletype clock 1, or the IOT 6421 did not skip after an interrupt from clock 1.

##### 4.5.2 Address 0307 Data Error

The processor halts with a line number displayed in the AC indicating that a data error has occurred on the line. Press CONTINUE; the processor halts at address 0312 with the character that should have been received in the AC. Press CONTINUE again, and the processor halts at address 0315 with the character that was received in the AC.

##### 4.5.3 Address 1055 Data Error (Output AC Switches Routine)

Processor halts with the character transmitted in the AC. This could be a carriage return, a line feed, or the code contained in the SWITCH REGISTER. Press CONTINUE, and the processor halts at address 1061 with the character that was received displayed in the AC.

##### 4.5.4 Detection of an Open Input Line

If the data error halt at address 0312 is consistently executed with a specific line number and the character received is always 0, it indicates that the input line is in a constant "SPACE" or open line state.

##### 4.5.5 Detection of a Constant "MARK"

The 680 Character Exerciser was written with the assumption that the Teletype lines being tested are full duplex. Also, the program assumes that any input received during the concurrent output test is due to the fact that the line outputs are tied to the line inputs. Therefore, if there is no input on a line or any of the lines the program does not consider it an error. Assuming the test is operating with inputs tied to outputs, the only way to be sure that a line is transmitting and receiving is to open the input line in order to force a data error.

Note that this procedure is likely to show up any errors in translating line numbers; i.e., input line 17 is opened and the processor does not halt. The line then is closed again and an error is indicated, for example, on line 37. The malfunction is probably in the decoding of line 17.

#### 4.6 Recovery From Such Errors

Pressing CONTINUE causes the program to proceed. It is possible that an error halt will leave some of the output lines in a "SPACE" condition. If this happens, any Teletypes that are tied to these outputs will run free until the program continues. Also, in the process of opening lines to test for transmitting and receiving, it is possible to get the input of a line out of sequence with the output. In this case, it is necessary to restart the test.

#### 5. RESTRICTIONS

The Character Assembly Subroutines scan the lines in even multiples of eight lines. Therefore, it is possible that an error could be indicated on a line or lines that have not been selected to be run, e.g., in the Output Switch Register Routine, if line 1 is selected to be run, the inputs of lines 2, 3, 4, 5, 6, 7, and 10<sub>8</sub> should be placed in the "MARK" state.

#### 6. DESCRIPTION

##### 6.1 Discussion

##### 6.1.1 General

The 680 5-Bit Character Exerciser transmits and/or receives 5-bit coded Teletype data at the line speed of Teletype clock 1 and will verify correct operation of the 680 DCS hardware. The program allows parameters to exercise up to 128 lines and includes three test modes: Concurrent Output, Concurrent Input, and Output the Switch Register.

##### 6.1.2 Test Descriptions

##### 6.1.2.1 Concurrent Output Test

The Concurrent Output Test accepts parameters to transmit on from 1 through 128 lines. The program assumes that all the lines selected are full duplex and that any input is due to an output line jumpered to an input line.

The following message is transmitted on all lines selected:

```
THE QUICK BROWN FOX JUMPS OVER THE LAZY DOG
0123456789 -?:$!&#'( ).,;/ "
RYRYRY
```

If input is received on any of the lines selected, the program verifies that the characters received are received in the same order they are transmitted. Also, since the input is tested almost independently of the output, it is possible to jumper input lines to output lines in any configuration that is desired, including and/or other than 1 for 1.

##### 6.1.2.2 Concurrent Input Test

The Concurrent Input Test initializes to scan all 128 lines for input. When a character is received on a line, the program transmits the character received on the same line.

##### 6.1.2.3 Output The Switch Register Routine

This Routine transmits the code contained in SWITCH REGISTER bits 7 through 11 on the line specified by the contents of the SWITCH REGISTER when the routine is started. The program transmits a carriage return, line feed, the contents of the SWITCH REGISTER 64 times, and then repeats. If SWITCH REGISTER bit 0 is a 1, the program transmits a character, waits for it to be input and then verifies that the output and input are equal. Since the input is two characters behind the output (due to the double buffering of output characters in the character assembly subroutines), the compare mode must be selected at the time the processor halts at address 1002.

7. METHODS (Not Applicable)
8. FORMAT (Not Applicable)
9. EXECUTION TIME (Not Applicable)
10. PROGRAM
- 10.4 Program Listing

PAUSE

/TYPE 680 TELETYPE LINE MULTIPLEXER 5 BIT  
/EXERCISER  
/KFN 10/15/65

/START AT 200  
/WITH STARTING LINE NUMBER IN SWITCHES  
/COMPUTER WILL HALT  
/SET SWITCHES = TO NUMBER OF LINES  
/CONTINUE  
/PROGRAM WILL RUN UNTIL AN ERROR  
/IS DETECTED OR UNTIL STOP

/INTERRUPT ROUTINE

```

*0001
0001 6421 INRUPT,      T5SKP          /WAS IT 5 BIT CLOCK
0002 7402      HLT          /NO, UNKNOWN INTERRUPT
0003 5404      JMP I .+1      /YES, PROCESS INTERRUPT
0004 5200      T5DIS          /TT5BGN

*200
0200 7604 DCSTST,      CLA OSR          /GET STARTING LINE
0201 3230      DCA STRLIN          /SAVE IT
0202 7402      HLT          /HALT, WAIT NUMBER OF LINES
0203 7604      CLA OSR          /GET NO OF LINES
0204 3331      DCA NUMLIN          /SAVE IT
0205 1331      TAD NUMLIN          /NO LINES
0206 7041      CIA          /2'S COMPLEMENT
0207 3332      DCA LINDEX          /SAVE FOR COUNTING
0210 1332      TAD LINDEX          /SET UP
0211 3345      DCA NDEX          /INDICES
0212 1333      TAD OUTDEX          /FOR INITIAL
0213 1230      TAD STRLIN          /SET UP OF
0214 3011      DCA Z 11          /INPUT AND
0215 1335      TAD INPDEX          /OUTPUT BUFFERS
0216 1230      TAD STRLIN
0217 3012      DCA Z 12
0220 1337 DCRFLP,      TAD CHARAC          /ADDRESS OF CHAR TABLE
0221 3411      DCA I Z 11          /TO OUTPUT BUFFER
0222 1337      TAD CHARAC          /ADDRESS TO COMPARE
0223 3412      DCA I Z 12          /CHAR TO INPUT BUFFER
0224 2345      ISZ NDEX          /GOT ALL LINES
0225 5220      JMP DCRFLP          /NO
0226 1331      TAD NUMLIN          /SET NO OF LINES IN AC
0227 4552      T5INIT          /INITIALIZE TTY
0230 0000 STRLIN,      0          /STARTING LINE NUMBER

```

0231	6424	TT50N	/TURN CLOCK ON
0232	6001	ION	/TURN INTERRUPTS ON
0233	7200	CLA	
0234	1230	TAD STRLIN	/FORM START OF
0235	1334	TAD OUTDEX+1	/OUTPUT BUFFER
0236	3340	DCA CUROUT	
0237	1230	TAD STRLIN	/GET FIRST LINE NO.
0240	3341	DCA CURLIN	
0241	1332	TAD LINDEFX	
0242	3345	DCA NDEX	
0243	1740	OTLOOP, TAD I CUROUT	/GET POINTER FOR NEXT OUTPUT
0244	3342	DCA SAVCUR	
0245	1742	TAD I SAVCUR	/GET NEXT OUTPUT CHARACTER
0246	3177	DCA Z TTCHAR	
0247	1341	TAD CURLIN	/GET LINE NUMBER
0250	4550	T5SOF	/CHECK FOR FREE OUTPUT
0251	7410	SKP	/OUTPUT NOT FREE
0252	2740	ISZ I CUROUT	/ADD 1 TO CHARACTER POINTER
0253	1740	TAD I CUROUT	
0254	7041	CIA	/HAS CURRENT
0255	1343	TAD LSTCHR	/LINE REACHED
0256	7440	SZA	/THE END OF OUTPUT
0257	5262	JMP .+3	/NO
0260	1337	TAD CHARAC	/YES RESET POINTER
0261	3740	DCA I CUROUT	/TO FIRST CHARACTER
0262	2340	ISZ CUROUT	/ADVANCE TO NEXT LINE
0263	2341	ISZ CURLIN	/ADVANCE TO NEXT LINE
0264	7200	CLA	/CLEAR FOR TAD
0265	2345	ISZ NDEX	/TESTED ALL LINED FOR FREE
0266	5243	JMP OTLOOP	/NO, TRY NEXT LINE
0267	4551	INLOOP, T5SIR	/ANY INPUT AVAILABLE
0270	5233	JMP STRLIN+3	/NO, OUTPUT AGAIN
0271	3341	DCA CURLIN	/YES, SAVE LINE NO.
0272	1341	TAD CURLIN	
0273	1336	TAD INPDFX+1	/FIND INPUT POINTER
0274	3340	DCA CUROUT	/FOR THE LINE
0275	1740	TAD I CUROUT	/GET INPUT POINTER
0276	3342	DCA SAVCUR	
0277	1177	TAD TTCHAR	/GET NEXT INPUT CHARACTER
0300	0344	AND LSTCHR+1	/37
0301	7041	CIA	
0302	1742	TAD I SAVCUR	/CHARACTER INPUT
0303	7450	SNA	/EQUAL CHARACTER EXPECTED
0304	5317	JMP INCINP	/YES, INCREMENT POINTER
0305	7200	CLA	
0306	1341	TAD CURLIN	/LINE NUMBER TO AC
0307	7402	HLT	/HALT
0310	7200	CLA	
0311	1742	TAD I SAVCUR	/CHARACTER EXPECTED TO AC
0312	7402	HLT	/HALT
0313	7200	CLA	
0314	1177	TAD Z TTCHAR	/CHARACTER RECEIVED
0315	0344	AND LSTCHR+1	
0316	7402	HLT	/HALT
0317	2740	INCINP, ISZ I CUROUT	/ADVANCE INPUT POINTER
0320	7200	CLA	
0321	1740	TAD I CUROUT	/GET INCREMENTED POINTER



0322	7041	CIA		
0323	1343	TAD LSTCHR		/LINE AT END OF
0324	7440	SZA		/INPUT CHARACTERS
0325	5267	JMP INLOOP		/NO, TEST FOR MORE INPUT
0326	1337	TAD CHARAC		/RESET INPUT POINTER
0327	3740	DCA I CUROUT		/TO FIRST CHARACTER
0330	5267	JMP INLOOP		/TEST FOR MORE INPUT
0331	0000	NUMLIN,	0	/NUMBER OF LINES
0332	0000	LINDEX,	0	/2'S COMPLEMENT NO OF LINES
0333	2177	OUTDEX,	BUFR-1	/FOR FILLING OUTPUT BUFFER
0334	2200	BUFR		/FOR CHECKING OUTPUT BUFFER
0335	2377	INPDEX,	INBUFR-1	/FOR FILLING INPUT BUFFER
0336	2400	INBUFR		
0337	0400	CHARAC,	CODES	/START OF 5 BIT CODES
0340	0000	CUROUT,	0	/TEMP STORAGE BUFFER POINTER
0341	0000	CURLIN,	0	/TEMP STORAGE CURRENT LINE NO.
0342	0000	SAVCUR,	0	/TEMP STORAGE CHARACTER POINTER
0343	0527	LSTCHR,	NDCOD+1	/FOR DETECTING END OF CHARACTERS
0344	0037	37		/FOR ANDING TICHAR
0345	0000	NDEX,	0	/FOR COUTING

BUFR=2200

INBUFR=2400

/5 BIT CODES OUTPUT BY DCSTST

\*400

0400	0010	CODES,	10	/CARRIAGE RETURN
0401	0002	02		/LINE FEED
0402	0002	02		/LINE FEED
0403	0037	37		/LETTERS
0404	0020	20		/T
0405	0024	24		/H
0406	0001	01		/E
0407	0004	04		/SPACE
0410	0027	27		/Q
0411	0007	07		/U
0412	0006	06		/I
0413	0016	16		/C
0414	0017	17		/K
0415	0004	04		/SPACE
0416	0031	31		/B
0417	0012	12		/R
0420	0030	30		/O
0421	0023	23		/W
0422	0014	14		/N
0423	0004	04		/SPACE
0424	0015	15		/F
0425	0030	30		/O
0426	0035	35		/X
0427	0004	04		/SPACE
0433	0013	13		/J
0431	0007	07		/U
0432	0034	34		/M
0433	0026	26		/P
0434	0005	05		/S
0435	0004	04		/SPACE
0436	0030	30		/O

0437	0036	36	/V
0440	0001	01	/E
0441	0012	12	/R
0442	0004	04	/SPACE
0443	0020	20	/T
0444	0024	24	/H
0445	0001	01	/E
0446	0004	04	/SPACE
0447	0022	22	/L
0450	0003	03	/A
0451	0021	21	/Z
0452	0025	25	/Y
0453	0004	04	/SPACE
0454	0011	11	/D
0455	0030	30	/O
0456	0032	32	/G
0457	0010	10	/CARRIAGE RETURN
0460	0002	02	/LINE FEED
0461	0033	33	/FIGURES
0462	0026	26	/0
0463	0027	27	/1
0464	0023	23	/2
0465	0001	01	/3
0466	0012	12	/4
0467	0020	20	/5
0470	0025	25	/6
0471	0007	07	/7
0472	0006	06	/9
0473	0030	30	/9
0474	0004	04	/SPACE
0475	0003	03	/MINUS
0476	0031	31	/QUESTION
0477	0016	16	/COLON
0500	0011	11	/DOLLAR
0501	0015	15	/EXCLAMATION
0502	0032	32	/AND
0503	0024	24	/NUMBER SIGN
0504	0013	13	/APOSTHROPHE
0505	0017	17	/LEFT PARENS
0506	0022	22	/RIGHT PARENS
0507	0034	34	/PERIOD
0510	0014	14	/COMMA
0511	0036	36	/SEMI COLON
0512	0035	35	/SLASH
0513	0021	21	/QUOTES
0514	0010	10	/CARRIAGE RETURN
0515	0002	02	/LINE FEED
0516	0037	37	/LETTERS
0517	0012	12	/R
0520	0025	25	/Y
0521	0012	12	/R
0522	0025	25	/Y
0523	0012	12	/R
0524	0025	25	/Y
0525	0012	12	/R
0526	0025	NDCOD,	/Y

```

/CHARACTER ECHO
/START AT 600
/PROGRAM INITIALIZES FOR 128 LINES
/PROGRAM WAITS FOR INPUTS
/TRANSMITS
/THE CHARACTERS RECEIVED
/THEN WAITS FOR MORE INPUTS
/PROGRAM RUNS UNTIL "STOP"

```

```

/ECHO
/INPUT CHARACTERS
/THEN OUTPUT THEM

```

```

*600
0600 7200 ECHOTS,      CLA           /GET LINE NUMBER
0601 1216      TAD ONE28          /AC=128
0602 4552      T5INIT             /SETUP FOR 128 LINES
0603 0000 ECHLIN,      0           /LINE NUMBER 0 STARTING LINE
0604 6424      TT5ON              /CLOCK ON
0605 6001      ION                /ENABLE INTERRUPTS
0606 4551      T5SIR              /SKIP IF INPUT
0607 5206      JMP .-1            /INPUT NOT READY, WAIT
0610 3215      DCA FCHDEX         /SAVE LINE NUMBER
0611 1215      TAD ECHDEX         /GET LINE NUMBER
0612 4550      T5SOF              /OUTPUT
0613 5211      JMP .-2            /NOT FREE TRY AGAIN
0614 5206      JMP FCHLIN+3       /WAIT MORE INPUT
0615 0000 ECHDEX,      0

```

```

0616 0200 ONE28,      0200
/TRANSMIT AC SWITCHES
/START AT 1000
/WITH SWITCHFS = LINE NUMBER
/COMPUTER WILL HALT
/SW0 = 0 COMPUTER WILL NOT COMPARE INPUT
/SW0 = 1 COMPUTER COMPARES INPUT AGAINST OUTPUT
/SW7 TO SW11 = CHARACTER TRANSMITTED
/SW0 MAY NOT BE ALTERED ONCE RUNNING
/SW7 TO SW11 MAY BE ALTERED
/PROGRAM TRANSMITS CARRIAGE RETURN/LINE FEED
/THE SW7 TO SW11 64 TIMES
/THEN CARRIAGE RETURN LINE FEED/64 CHARACTERS
/PROGRAM RUNS UNTIL "STOP"

```

```

/TRANSMIT AC SWITCHES

```

```

*1000
1000 7604 TRANSR,      CLA OSR           /GET LINE NO
1001 3205      DCA TRALIN          /T5INIT+1
1002 7402      HLT                /WAIT FOR CHARACTER
1003 7001      IAC                /AC=NO OF LINES (1)
1004 4552      T5INIT             /INITIALIZE TTY ROUTINES
1005 0000 TRALIN,      0           /LINE NUMBER
1006 6424      TT5ON              /TURN CLOCK ON
1007 6001      ION                /ENABLE INTERRUPTS
1010 1264      TAD K64            /-64
1011 3266      DCA TRADFX         /FOR COUNTING
1012 1267      TAD TRADFX+1       /CARRIAGE RETURN

```

```

1013 3271 DCA TRACHR /OUTPUT CARRIAGE RETURN
1014 4227 JMS TRAOUT /OUTPUT
1015 1270 TAD TRADEX+2 /LINE FEED
1016 3271 DCA TRACHR /OUTPUT CHARACTERS
1017 4227 JMS TRAOUT /OUTPUT
1020 7604 CLA OSR /GET CHARACTER IN SWITCHES
1021 0265 AND THREF7 /CLEAR ALL BUT LWR 5
1022 3271 DCA TRACHR /OUTPUT CHARACTER
1023 4227 JMS TRAOUT /OUTPUT
1024 2266 ISZ TRADEX /OUTPUT 64 CHARACTERS
1025 5220 JMP .-5 /TEST SWITCHES AGAIN
1026 5210 JMP TRALIN+3 /OUTPUT CAR. RET./LINE FEED
1027 5227 TRAOUT, JMP .
1030 7200 CLA
1031 1271 TAD TRACHR /GET OUTPUT CHARACTER
1032 3177 DCA TTCHAR /FOR OUTPUTTING
1033 1205 TAD TRALIN /LINE NUMBER
1034 4550 T5SOF
1035 5230 JMP .-5 /OUTPUT NOT FREE, TRY AGAIN
1036 7604 TRASW0, CLA OSR /GET SWITCHES
1037 7004 RAL /BIT 0 TO LINK
1040 7620 SNL CLA /0 SET
1041 5627 JMP I TRAOUT /NO, EXIT

1042 4551 TRAWAT, T5SIR /SKIP IF INPUT
1043 5242 JMP .-1 /NO INPUT, TEST SW0
1044 7200 CLA
1045 1177 TAD TTCHAR /GET CHARACTER
1046 0265 AND THREF7 /CLEAR STOP BIT
1047 7041 CMA IAC
1050 1271 TAD TRACHR /CHARACTER RECEIVED=SENT
1051 7450 SNA
1052 5627 JMP I TRAOUT /EXIT
1053 7200 CLA
1054 1271 TAD TRACHR /CHARACTER SENT
1055 7402 HLT /HALT
1056 7200 CLA
1057 1177 TAD TTCHAR /CHARACTER RECEIVED
1060 0265 AND THREF7 /TO LAST 5 BITS
1061 7402 HLT /HLT
1062 7200 CLA
1063 5627 JMP I TRAOUT /EXIT

1064 7700 K64, 7700 /-64
1065 0037 THREE7, 37 /FOR ANDING
1066 0000 TRADEX, 0 /FOR COUNTING
1067 0010 10 /CARRIAGE RETURN
1070 0002 02 /LINE FEED
1071 0000 TRACHR, 0 /CHARACTER STORAGE
/TYPE 680 TELETYPE LINE MULTIPLEXER
/CHARACTER ASSEMBLY ROUTINE
/LMH 910/15/65 5 BIT

TTI=6402 /TELETYPE INPUT COMMAND
TTO=6404 /TELETYPE OUTPUT COMMAND
TTCL=6411 /CLEAR LINE REGISTER
TTRL=6414 /READ LINE REGISTER
TTSL=6412 /SET LINE REGISTER, CLR AC

```

TT50N=6424           /TURN CLOCK ON  
TT50FF=6422         /TURN CLOCK OFF  
T5SKP=6421          /SKIP ON CLOCK FLAG  
TTINCR=6401         /INCREMENT LINE REGISTER

/680 LINE MULTIPLEXER  
/LIST OF ITEMS  
T5IRF=7200  
T50BF2=7000  
T50BF=6600  
T5IN=5600  
TT5PG0=145  
T5AX1=10  
T5AX2=11  
T5AX3=12  
T5AX4=13  
TT5RGN=5200  
TTCHAR=177

```

*TT5PG0
0145 0000 T5INFL, 0 /INPUT READY FLAG
0146 7177 T5RFK, T5IRF-1 /TO RESET INPUT BUFFER POINTER
0147 0000 T5NL, 0 /-NUMBER OF LINES
0150 5400 T5SOUT, T5OUTS /SKIP IF OUTPUT FREE
0151 5423 T5SIN, T5INS /SKIP IF INPUT READY
0152 5447 T5G0, T5G0S /INITIALIZE ROUTINE
0153 6600 T5OUTK, T50BF /POINTER TO 1ST OUTPUT BUFFER
0154 7774 T5CNT1, -4 /HOLDS MAJOR LOOP COUNTER
0155 0000 T5CNT2, 0 /MINOR LOOP COUNTER
0156 0000 T5CNT3, 0 /COUNTER FOR INPUT BUFFER
0157 0177 T5K10, 177 /FOR ANDING
0160 7000 T5K36, T50BF2 /2ND OUTPUT BUFFER
0161 0000 T5CNT5, 0 /OUTPUT COUNTER
0162 0000 T5CNT6, 0 /7 BIT COUNTER
0163 7770 T5K2, -10 /TO RESET BIT COUNTER
0164 5600 T5K3, T5IN /RESET INPUT TTI POINTER
0165 7776 T5K5, -2 /FOR SUBTRACTION
0166 0400 T5K6, 400 /TO RESET 5 BIT ASSEMBLY WORD
0167 6600 T5K7, T50BF /K FOR 1ST OUTPUT BUFFER
0170 5221 T5K8, T5COM /TO ENTER COMMON ROUTINE
0171 0000 T5K9, 0 /LINE NUMBER -1
0172 6577 T5K9A, T50RF-1 /FOR CLEARING
0173 5237 T5K9R, T5CM1A /TO AVOID OUTPUTTING
0174 5361 T5K9C, JMP T5CM10 /TO SET OUTPUT BUFFER FROM DOUBLE BUFFER
0175 5221 T5K9D, T5COM /FOR NORMAL RETURN
0176 7000 T5K9F, NOP /TO DO INPUT ONLY

*TT5RGN
/MULTIPLE LEVEL INTERRUPT ROUTINE
/ALLOWS MULTIPLE LEVEL INTERRUPT TO THIS ROUTINE
/AND UNLIMITED

5200 2366 T5DIS, T5Z T5LC /LEVEL COUNTER
5201 5216 JMP T5DIS3 /2ND LEVEL INTERRUPT
5202 3367 DCA T5SA /SAVE ACCUMULATOR
5203 7010 RAR /GET LINK
5204 3370 DCA T5SVLK /SAVE LINK
5205 1000 TAD Z 0 /INTERRUPT ADDRESS

```

```

5206 3371      DCA T5SV0      /SAVE ADDRESS
5207 6414      TTRL          /READ LINE NUMBER
5210 3372      DCA T5SVLN     /SAVE LINE NUMBER
5211 6424      TT50N         /TO CLEAR CLOCK FLAG ONLY
5212 6001      T5DIS2, ION    /RE-ENABLE PROGRAM INTERRUPT
5213 1171      TAD T5K9       /STARTING LINE-1
5214 6413      TTSL+1        /SFT LINE REGISTER, CLR AC
5215 5564      JMP I Z T5K3   /SET LINE REGISTER, CLR AC

/2ND LEVEL INTERRUPT
5216 6424      T5DIS3, TT50N  /CLEAR CLOCK FLAG
5217 6001      ION          /RE-ENABLE PROGRAM INTERRUPT
5220 5400      JMP I Z 0      /RETURN TO THE MAIN PROGRAM

/RETURN FROM INPUT TTI LOOP
5221 1373      T5COM, TAD T5MNC /-NO. OF LINES/8
5222 3155      DCA Z T5CNT2    /MINOR LOOP COUNTER
5223 1375      TAD T5LN        /LINE NUMBER
5224 6413      TTSL+1        /SET LINE NUMBER
5225 1553      T5COM0, TAD I Z T5OUTK /OUTPUT WORD
5226 7450      SNA           /CHARACTER AVAILABLE
5227 5351      JMP T5COM8      /NOTHING TO TRANSMIT
5230 6405      TIO+1         /INCREMENT AND TRANSMIT
5231 3553      DCA I Z T5OUTK  /RESTORE CHARACTER
5232 2153      T5COM1, ISZ Z T5OUTK /UPDATE OUTPUT POINTER
5233 2155      ISZ Z T5CNT2    /ARE ONE-EIGHTH OF LINES DONE
5234 5225      JMP T5COM0      /CHECK NEXT OUTPUT LINE
5235 6414      TTRL          /READ LINE NUMBER
5236 3375      DCA T5LN        /SAVE LINE NUMBER
5237 1374      T5CM1A, TAD T5MNC2 /NO OF LINES/4
5240 3155      DCA T5CNT2     /MINOR LOOP COUNTER
5241 2010      T5COM2, ISZ T5AX1 /ADVANCE FOR NEXT INPUT LINE
5242 1410      TAD I Z T5AX1   /CHARACTER ASSEMBLY WORD
5243 7112      CLL RTR        /PUT BIT 10 IN LINK
5244 7430      SZL           /CHARACTER NOT COMPLETED
5245 5326      JMP T5COM6      /STORE CHARACTER
5246 7200      CLA           /CLEAR AC FOR TAD
5247 7000      T5COM3, NOP     /OR JMP T5CM10
5250 2010      ISZ Z T5AX1     /UPDATE FOR NEXT INPUT LINE
5251 2376      ISZ T5LN2      /UPDATE LINE NUMBER
5252 2155      ISZ T5CNT2     /ARE ONE-FOURTH OF LINES CHECKED?
5253 5241      JMP T5COM2     /CHECK NEXT LINE
5254 2154      T5COM4, ISZ T5CNT1 /HAVE ALL INPUT LINES BEEN CHECKED
5255 5310      JMP T5COM5     /RESET AND DISMISS
5256 1164      TAD Z T5K3     /T5IN
5257 3010      DCA Z T5AX1    /RESET TTI POINTER
5260 1171      TAD Z T5K9     /START LINE-1
5261 7001      IAC          /SET TO FIRST LINE
5262 3376      DCA T5LN2      /RESET LINE NUMBER
5263 1377      TAD T5K2A      /-4
5264 3154      DCA T5CNT1     /INPUT CHECK COUNTER
5265 2161      ISZ Z T5CNT5    /HAVE ALL OUTPUT LINES BEEN CHECKED
5266 5310      JMP T5COM5     /RESET AND DISMISS
5267 1165      TAD Z T5K5     /-2
5270 3161      DCA Z T5CNT5    /RESET COUNTER
5271 1171      TAD Z T5K9     /START LINE-1
5272 3375      DCA T5LN        /RESET LINE NUMBER

```

5273	2162	ISZ Z T5CNT6	/ENDING 7TH BIT?
5274	5353	JMP T5COM9	/NO RESET NORMALLY
5275	1163	TAD T5K2	/-10
5276	3162	DCA Z T5CNT6	/RESET COUNTER
5277	2161	ISZ Z T5CNT5	/ADD 1 TO COUNTER
5300	1172	TAD Z T5K9A	/T50BF-1
5301	3013	DCA Z T5AX4	/OUTPUT POINTER
5302	1160	TAD Z T5K36	/T50BF2
5303	3153	DCA T5OUTK	/2ND BUFFER POINTER
5304	1173	TAD Z T5K9B	/SPECIAL ADDRESS, T5CM1A
5305	3170	DCA Z T5K8	/RESET ADDRESS
5306	1174	TAD Z T5K9C	/JMP T5CM10
5307	3247	DCA T5COM3	/SET TO DO OUTPUT
5310	6002	T5COM5, IOF	/TURN OFF INTERRUPT
5311	7240	STA	/-1
5312	1366	TAD T5LC	/LEVEL COUNTER
5313	3366	DCA T5LC	/RESTORE LEVEL COUNTER
5314	1366	TAD T5LC	/LEVEL COUNTER
5315	7700	SMA CLA	/RESTORE AC, ETC.
5316	5212	JMP T5DIS2	/CHECK INPUT AGAIN, ETC.
5317	1372	TAD T5SVLN	/LINE NUMBER
5320	6413	TTSL+1	/SET LINE REGISTER, CLR AC
5321	1370	TAD T5SVLK	/PICK UP LINK
5322	7104	CLL RAL	/RESTORE LINK
5323	1367	TAD T5SA	/RESTORE AC
5324	6001	ION	/RE-ENABLE PROGRAM INTERRUPT
5325	5771	JMP I T5SV0	/RETURN TO THE MAIN PROGRAM
5326	7112	T5COM6, CLL RTR	/REMOVE START CODE
5327	7012	RTR	
5330	3411	DCA I Z T5AX2	/STORE CHARACTER
5331	1376	TAD T5LN2	/LINE NUMBER
5332	3411	DCA I Z T5AX2	/STORE LINE NUMBER
5333	1010	TAD Z T5AX1	/TTI POINTER
5334	1165	TAD Z T5K5	/-2
5335	3010	DCA Z T5AX1	/RESET POINTER
5336	3410	DCA I Z T5AX1	/ZERO STATUS AND COUNTER WORD
5337	1166	TAD Z T5K6	/WORD TO RESTORE ASSEMBLY WORD
5340	3410	DCA I Z T5AX1	/RESET CHARACTER ASSEMBLY WORD
5341	2145	ISZ Z T5INFL	/SET INPUT READY FLAG
5342	2156	ISZ Z T5CNT3	/HAS END OF BUFFER BEEN REACHED?
5343	5247	JMP T5COM3	/CONTINUE
5344	1146	T5COM7, TAD Z T5BFX	/T5IRF-1
5345	3011	DCA Z T5AX2	/RESET INPUT BUFFER ADDRESS
5346	1147	TAD T5NL	/--UMBER OF LINES
5347	3156	DCA Z T5CNT3	/RESET LENGTH COUNTER
5350	5247	JMP T5COM3	/CONTINUE
5351	6401	T5COM8, TTINCR	/INCREMENT LINE NUMBER
5352	5232	JMP T5COM1	/CONTINUE
5353	1167	T5COM9, TAD Z T5K7	/T50BF
5354	3153	DCA Z T5OUTK	/RESET OUTPUT POINTER
5355	1175	TAD Z T5K9D	/NORMAL ADDRESS, T5COM
5356	3170	DCA Z T5K8	/RESET ADDRESS
5357	1176	TAD Z T5K9E	/NOP
5360	5307	JMP T5COM5-1	/CONTINUE
5361	1553	T5CM10, TAD I Z T5OUTK	/2ND BUFFER CHARACTER
5362	3413	DCA I Z T5AX4	/STORE IN 1ST BUFFER
5363	3553	DCA I T5OUTK	/CLEAR 2ND BUFFER

```

5364 2153      ISZ T5OUTK      /UPDATE POINTER
5365 5250      JMP T5COM3+1    /CONTINUE

/CONSTANTS
5366 7777      T5LC,          -1      /INTERRUPT LEVEL COUNTER
5367 0000      T5SA,           0      /SAVE ACCUMULATOR
5370 0000      T5SVLK,         0      /SAVE LINK
5371 0000      T5SV0,          0      /SAVE PROGRAM COUNTER
5372 0000      T5SVLN,         0      /SAVE LINE NUMBER
5373 0000      T5MNC,          0      /-NO OF LINES/8
5374 0000      T5MNC2,         0      /-NO OF LINES/4
5375 0000      T5LN,           0      /LINE NUMBER FOR OUTPUT
5376 0000      T5LN2,          0      /LINE NUMBER FOR INPUT
5377 7774      T5K2A,          -4      /TO RESET MAJOR LOOP COUNTER

/PSEUDO-OPERATIONS
*TT5RGN+200
/SKIP IF OUTPUT IS FREE AND TRANSMIT CHARACTER AT TTCHAR, OTHERWISE
/DOON'T SKIP LINE NUMRER MUST BE IN AC. 24US MIN., 42US MAX.
T5SOF=JMS I Z T5SOUT
5400 0000      T5OUTS,         0
5401 0157      AND Z T5K10      /177
5402 1217      TAD T5SL        /-STARTING LINE NUMBER
5403 1160      TAD Z T5K36      /OUTPUT BUFFER ADDRESS
5404 3220      DCA T5WA        /WORK AREA
5405 1620      TAD I T5WA      /OUTPUT CHARACTER
5406 7640      SZA CLA        /SKIP IF FREE
5407 5600      JMP I T5OUTS     /EXIT
5410 1177      TAD Z TTCHAR    /PICK UP CHARACTER
5411 0221      AND T5K11       /5 BITS ONLY
5412 1222      TAD T5K12       /140 FOR STOP CODE
5413 7104      CLL RAL        /CREATE START CODE
5414 3620      DCA I T5WA      /STORE CHARACTER IN TABLE
5415 2200      ISZ T5OUTS      /INDEX EXIT
5416 5600      JMP I T5OUTS     /EXIT
5417 0000      T5SL,           0      /-STARTING LINE NUMBER
5420 0000      T5WA,           0      /WORK AREA
5421 0037      T5K11,          37      /FOR 5 BIT CODE
5422 0040      T5K12,          40      /FOR STOP CODE

/SKIP IF CHARACTER AVAILABLE AND RFTURN WITH LINE NUMBER IN AC
/CHAR AT TTCHAR. OTHERWISE DO NOT SKIP
/15US MIN., 48US MAX., 37.5US NORMAL IF READY
T5SIR=JMS I Z T5SIN
5423 0000      T5INS,          0
5424 6002      IOF
5425 7240      CLA CMA        /SET AC FOR TAD-1
5426 1145      TAD Z T5INFL     /INPUT FLAG COUNTER-1
5427 7510      SPA            /SOMETHING AVAILABLE
5430 5244      JMP T5INON      /EXIT
5431 3145      DCA Z T5INFL     /RESTORE FLAG COUNTER
5432 2246      ISZ T5CNT4      /END OF BUFFER? START AT -N-1
5433 5240      JMP .+5         /GET CHARACTER
5434 1147      TAD Z T5NL      /-NUMBER OF LINES
5435 3246      DCA T5CNT4      /RESET COUNTER
5436 1146      TAD Z T5RFK     /BUFFER ADDRESS-1
5437 3012      DCA Z T5AX3     /RESET ADDRESS
5440 1412      TAD I Z T5AX3   /PICK UP CHARACTER
5441 3177      DCA Z TTCHAR    /STORE CHARACTER

```



5442	1412	TAD I Z T5AX3	/PICK UP LINE NO.
5443	2223	ISZ T5INS	/INDEX EXIT
5444	6001	T5INON, ION	
5445	5623	JMP I T5INS	/EXIT
5446	0000	T5CNT4, 0	/-NUMBER OF LINES
		/INITIALIZATION ROUTINE	
		/ENTER WITH NUMBER OF LINES IN AC	
		/FORMAT T5INIT	
		/ 1ST LINE NO.	
		T5INIT=JMS I Z T5GO	
5447	0000	T5GOS, 0	
5450	0355	AND T5K14	/377
5451	3147	DCA Z T5NL	/NO. OF LINES
5452	1147	TAD Z T5NL	/NO. OF LINES
5453	0356	AND T5K15	/7
5454	7640	SZA CLA	/MULTIPLE OF 8
5455	1357	TAD T5K16	/10
5456	1147	TAD Z T5NL	/NO. OF LINES
5457	0360	AND T5K17	/370
5460	7041	CIA	/TWO'S COMP. NUMBER OF LINES
5461	3147	DCA Z T5NL	/-N, CONSTANT
5462	1147	T5G01, TAD Z T5NL	/-N
5463	3156	DCA Z T5CNT3	/COUNTER
5464	1361	TAD T5K20	/T5IN-1
5465	3010	DCA Z T5AX1	/TO STORE TTI TABLE
5466	1362	TAD T5K21	/T50BF-1
5467	3011	DCA Z T5AX2	/TO CLEAR OUTPUT AREA
5470	1373	TAD T5K37	/T80BF2-1
5471	3012	DCA T5AX3	/TO CLEAR DOUBLE BUFFER
5472	1147	TAD Z T5NL	/-N
5473	3246	DCA T5CNT4	/FOR COUNTING
5474	1363	T5G02, TAD T5K22	/TTI+INCR
5475	3410	DCA I Z T5AX1	/STORE TTI
5476	3410	DCA I Z T5AX1	/CLEAR STATUS WORD
5477	1166	TAD Z T5K6	/ASSEMBLY RESET WORD
5500	3410	DCA I Z T5AX1	/RESET ASSEMBLY WORD
5501	3411	DCA I Z T5AX2	/ZERO OUTPUT WORD
5502	3412	DCA I Z T5AX3	/CLEAR DOUBLE BUFFER
5503	2246	ISZ T5CNT4	/COUNTER
5504	5274	JMP T5G02	/DO NEXT LINE
5505	1364	TAD T5K24	/JMP I Z T5K8
5506	3410	DCA I Z T5AX1	/STORE FINAL JUMP
5507	1147	TAD Z T5NL	/-N
5510	7012	RTR	/-N/4
5511	7010	RAR	/-N/8
5512	0365	AND T5K25	/17
5513	1366	TAD T5K26	/7760, MAKE NUMBER NEGATIVE
5514	3767	DCA I T5K27	/T5MNC
5515	1767	TAD I T5K27	/T5MNC
5516	1767	TAD I T5K27	/T5MNC -N/4
5517	3774	DCA I T5K38	/T5MNC2
5520	7240	STA	/-1

5521	3246	DCA T5CNT4	/SET CNTR TO SKIP 1ST TIME
5522	1146	TAD Z T5BFK	/T5IBF-1
5523	3011	DCA Z T5AX2	/SET INPUT BUFFER POINTER
5524	1370	TAD T5K28	/-4
5525	3154	DCA Z T5CNT1	/MAJOR LOOP COUNTER
5526	1165	TAD Z T5K5	/-2
5527	3161	DCA Z T5CNT5	/OUTPUT COUNTER
5530	1164	TAD Z T5K3	/T5IN+1
5531	3010	DCA Z T5AX1	/SET TTI POINTER
5532	1167	TAD T5K7	/T5ORF
5533	3153	DCA Z T5OUTK	/SET OUTPUT BUFFER POINTER
5534	7240	STA	/-1
5535	1647	TAD I T5G0S	/STARTING LINE NUMBER
5536	3171	DCA Z T5K9	/STARTING LINE NO-1
5537	1171	TAD Z T5K9	/STARTING LINE -1
5540	7040	CMA	/MAKE NEGATIVE
5541	3217	DCA T5SL	/-STARTING LINE NUMBER
5542	3145	DCA Z T5INFL	/CLEAR INPUT FLAG COUNTER
5543	7240	STA	/-1
5544	3771	DCA I T5K35	/T5LC, RESET INTERRUPT LEVEL COUNTER
5545	2247	ISZ T5G0S	/INDEX EXIT
5546	1372	TAD T5K35A	/-7
5547	3162	DCA Z T5CNT6	/SET SPECIAL 5-BIT COUNTER
5550	1175	TAD Z T5K9D	/T5COM
5551	3170	DCA Z T5K8	/TTI RETURN
5552	1176	TAD Z T5K9E	/NOP
5553	3775	DCA I T5K40	/T5COM3
5554	5647	JMP I T5G0S	/EXIT

## /CONSTANTS

5555	0377	T5K14,	377	/FOR ANDING
5556	0007	T5K15,	7	/FOR EVEN MULTIPLE OF 8
5557	0010	T5K16,	10	/FOR EVEN MULTIPLE OF 8
5560	0370	T5K17,	370	/FOR EVEN MULTIPLE OF 8
5561	5577	T5K20,	T5IN-1	/FOR STORING TTI'S
5562	6577	T5K21,	T5ORF-1	/FOR OUTPUT AREA
5563	6403	T5K22,	TTI+1	/TTI
5564	5570	T5K24,	JMP I Z T5K8	/FOR FINAL JUMP
5565	0017	T5K25,	17	/FOR -N/8
5566	7760	T5K26,	7760	/FOR MAKING NEGATIVE
5567	5373	T5K27,	T5MNC	/FOR -N/8
5570	7774	T5K28,	-4	/FOR MAJOR LOOP COUNTER
5571	5366	T5K35,	T5LC	/FOR INTERRUPT LEVEL COUNTER
5572	7771	T5K35A,	-7	/FOR 5-BIT COUNTER
5573	6777	T5K37,	T5ORF2-1	/FOR DOUBLE BUFFER
5574	5374	T5K38,	T5MNC2	/FOR -N/4
5575	5247	T5K40,	T5COM3	/FOR SWITCH

11. DIAGRAMS (Not Applicable)

12. REFERENCES

12.1 Other Library Programs

Digital-8-35-S-A

680 5-Bit Character Assembly Subroutines